


第 6 章 字模製作與目標比對



6-1 建立車牌字模圖檔

要知道影像目標到底是甚麼「字」？就必須建立標準答案的圖檔。本章將示範如何建立字模資料與比對字模的基本動作。首先是要製作標準的字模影像，如果你可以找到監理單位公布的標準字型圖檔當然最好，但是據我所知，台灣的七碼車牌是有公告標準字型，六碼的卻沒有！所以你只能找六碼的車牌影像，自己用影像編輯軟體如 PhotoShop 去切割製作出你要的個別字元的標準影像。

首先你要自訂標準字模的大小，台灣車牌的標準字型都是寬高比 1:2，本書設定的字寬為 25 點，字高為 50 點。當你切出字元影像之後，必須重新取樣，在**不鎖定長寬比**的條件下，讓圖檔變成 25x50 的大小。因為作字模比對時通常已是二值化影像，所以建議將影像二值化變成黑白圖，並以 BMP、GIF 或 PNG 的格式儲存。最不恰當的格式是 JPG，因為它的壓縮方式會讓字元輪廓變得模糊，不利於維持精準的字形，應該避免。

為了之後程式操作的方便，建議像上圖的方式依序命名，就是六碼字形 0-9，A-Z，接著是七碼字型的 0-9 與 A-Z，總共應該是 00-71 有 72 個字元圖檔。那為何最後又會多出 6 與 9 兩個字呢？原因是六碼車牌的 6 與 9 字常出現圓圈部分特別大的變形，導致用標準的 6 與 9 比對時符合度偏低，因此有必要加入這兩個變形字模。

6 9

這種字形例外的情況在車排辨識時不時會發生，本公司曾經承做過香港車牌的辨識，字形混亂的情況就任人咋舌！因為車牌畢竟不是監理單位自行開模生產製作的，有些國家地區甚至是車主自己找不同的廠商製作的！雖然監理單位應該都會公告字型與車牌大小比例等規範，但不保證每一家廠商都會嚴格遵守！當他們作的車牌略有不同，導致辨識軟體失誤時，車主與製造商是不會被開罰單的！

所以我們這些作辨識軟體的就必須有**例外處理的能力與機制**，不然要作出商業化產品就難了！簡單說，就是同一個字元在實際案例中，難免會出現筆畫位置明顯有差異的字型，此時我們只能為它們建立額外的字型，正常的 6 字模比對起來不像，還要用變形版的 6 再比對一次，如果符合的話答案也算 6，寫程式時就是變成例外的處理程序。

本書範例中刻意將此變形的 6 與 9 加入，就是希望大家學會如何在遭遇這種例外時可以彈性的增加字模，以提高辨識成功率！

6-2 字模建立與比對示範的專案



FolderBrowserDialog1 OpenFileDialog1 SaveFileDialog1

請開啟一個程式專案，建立如上的介面，上排左第一與第二個方框都是 `ListBox` 物件，用於雙層次選取字模資料顯示之用，第二個 `ListBox` 的 `MultiColumn` 屬性設為 `True`，

ColumnWidth 屬性設為 25 點，就可以呈現多行的畫面不必一直拉捲軸了！上排右方是兩個 PictureBox，分別用來顯示字模與等待比對的目標影像，我們還需要開關檔案與目錄的三個對話方塊 FolderBrowserDialog1、OpenFileDialog1 與 SaveFileDialog1 等。

這個專案首先將匯入我們製作的字元圖檔，就是「用影像載入字模」的那個按鍵功能，將其建立為二值化的 0 與 1 的陣列，1 代表目標，0 代表背景。接下來會「建立字模二進位檔案」就是將所有二值化陣列以位元組的方式集結成單一的二進位檔案匯出。

在此之後任何專案需要字模資料時，就不必再載入這麼多圖檔，直接將匯出的二進位檔案當作專案資源檔匯入專案即可。內行的讀者可能會問：怎麼不是用資料庫的方式處理？原因是字模資料量其實很小，如果刻意使用資料庫架構會降低執行效率，也會讓程式因為導入資料庫相關元件而變得較為笨重，直接以檔案匯入會較為精簡有效率。

接下來是「計算符合度」的功能計算，看看目標與任何一個字模的相似度，我們都會詳細示範如何實作

6-3 用影像載入字模

首先請宣告以下的全域變數：

```
Dim Pw As Integer = 25, Ph As Integer = 50 '字模的寬與高
Dim P(1, 35, Pw - 1, Ph - 1) As Byte '六與七碼車牌所有英數字二值化陣列
Dim P69(1, Pw - 1, Ph - 1) As Byte '變形的 6 與 9
Dim A(Pw - 1, Ph - 1) As Byte, B(Pw - 1, Ph - 1) As Byte '待比對的字模與目標二值化陣列
```

其中 P(0,35,x,y)是對應到 36 個標準六碼字型，P(1,35,x,y)是七碼字型，P69 則是兩個例外字型。前一節提到，字型例外是常見的意外，所以此地特別示範如何應對，如果你又發現了另一個常見的例外，可以自己新增一個字模，然後在此擴編 P69 陣列即可。接下來是載入圖檔將資訊寫入以上陣列的程式碼，就是「用影像載入字模」的功能按鍵，我們是將所有圖檔集中於一個目錄，所以是從選擇目錄開啟的動作為起始的。

```
'用影像載入字模
Private Sub Button1_Click(ByVal sender As Object, ByVal e As System.EventArgs) _
    If FolderBrowserDialog1.ShowDialog = Windows.Forms.DialogResult.OK Then
        Dim d As String = FolderBrowserDialog1.SelectedPath
        For k As Integer = 0 To 35
            '六碼字型
            Dim b6 As New Bitmap(d + "\" + Format(k, "00") + ".gif")
            For i As Integer = 0 To Pw - 1
                For j As Integer = 0 To Ph - 1
                    Dim C As Color = b6.GetPixel(i, j)
                    If C.R < 128 Then P(0, k, i, j) = 1 '黑點
                Next
            Next
            '七碼字型
```

```

Dim b7 As New Bitmap(d + "\" + Format(k + 36, "00") + ".gif")
For i As Integer = 0 To Pw - 1
    For j As Integer = 0 To Ph - 1
        Dim C As Color = b7.GetPixel(i, j)
        If C.R < 128 Then P(1, k, i, j) = 1 '黑點
    Next
Next
Next
For k As Integer = 72 To 73
    '六碼變形的 69 字型
    Dim b69 As New Bitmap(d + "\" + Format(k, "00") + ".gif")
    For i As Integer = 0 To Pw - 1
        For j As Integer = 0 To Ph - 1
            Dim C As Color = b69.GetPixel(i, j)
            If C.R < 128 Then P69(k - 72, i, j) = 1 '黑點
        Next
    Next
Next
End If
End Sub

```

基本上就是一一打開每一張圖檔，讀取每一個畫素，明顯偏黑的點就定義為字模陣列中的”1”值，其他的就是”0”了！此時我們當然想驗證一下載入是不是正確成功？就可以依序點選 ListBox1 與 ListBox2 來選擇顯示指定字模畫到 PictureBox1 上了！

```

'選擇字模
Private Sub ListBox1_SelectedIndexChanged(ByVal sender As Object, ByVal e As System.EventArgs) _
    Handles ListBox1.SelectedIndexChanged
    Dim knd As Integer = ListBox2.SelectedIndex '六或七碼
    If knd < 0 Then Exit Sub '未選擇
    Dim sC As Integer = ListBox1.SelectedIndex '選擇的字元
    If sC < 0 Then Exit Sub '未選擇
    If knd = 1 And sC > 35 Then Exit Sub '七碼車牌字元無變形的 69
    ReDim B(Pw - 1, Ph - 1) '待比對的字模二值化陣列
    For i As Integer = 0 To Pw - 1
        For j As Integer = 0 To Ph - 1
            If knd = 0 Then '六碼字型
                If sC <= 35 Then
                    B(i, j) = P(0, sC, i, j)
                Else
                    B(i, j) = P69(sC - 36, i, j)
                End If
            Else '七碼字型
                B(i, j) = P(1, sC, i, j)
            End If
        Next
    Next


```

Next

PictureBox1.Image = BWImg(B) '顯示字模影像

End Sub

這個過程其實就是將每個字模的二值化陣列重新繪製成影像，應該不難理解。執行畫面大致如下：



6-4 建立字模的二進位檔案

如果每次開啟車牌辨識軟體，都要匯入解讀這麼多圖檔，當然是很沒效率的，我們也不是真的需要它們以圖檔的方式呈現，只是需要那些二值化的陣列資料而已。所以合理的方式是將它們當作**資料庫**的概念，以資料檔案的方式一開啟程式就自動載入。首先必須先製作出包含所有字模二維陣列資料的二進位檔案，就是「**建立字模二進位檔案**」的按鍵功能，程式碼如下：

'建立字模二進位檔案

```
Private Sub Button2_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) _
```

```
Handles Button2.Click
```

```
If SaveFileDialog1.ShowDialog = Windows.Forms.DialogResult.OK Then
```

```
If My.Computer.FileSystem.FileExists(SaveFileDialog1.FileName) Then
```

```
My.Computer.FileSystem.DeleteFile(SaveFileDialog1.FileName)
```

```
End If
```

```
'寫出六與七碼字模
```

```
For m As Integer = 0 To 1
```

```
For k As Integer = 0 To 35
```

```
For j As Integer = 0 To Ph - 1
```

```
Dim x(Pw - 1) As Byte
```

```
For i As Integer = 0 To Pw - 1
```

```

        x(i) = P(m, k, i, j)
    Next
    My.Computer.FileSystem.WriteAllBytes(SaveFileDialog1.FileName, x, True)
Next
Next
Next
'寫出六碼變形 69 字模
For k As Integer = 0 To 1
    For j As Integer = 0 To Ph - 1
        Dim x(Pw - 1) As Byte
        For i As Integer = 0 To Pw - 1
            x(i) = P69(k, i, j)
        Next
        My.Computer.FileSystem.WriteAllBytes(SaveFileDialog1.FileName, x, True)
    Next
Next
End If
End Sub

```

6-5 以資源檔案方式匯入專案

不管你用甚麼檔名在上一節製作出一個檔案，請將此檔以**資源檔案**的方式加入本專案！在 My Project 專案屬性頁加入現有檔案，並將此資源命名為 font：



接下來在 Form_Load 事件中載入這個資源，為了方便後續專案使用，我們將載入動作獨立寫成一個副程式，程式碼如下：

```

'啟動程式載入字模
Private Sub Form1_Load(ByVal sender As Object, ByVal e As System.EventArgs) Handles MyBase.Load
    FontLoad()
End Sub
Private Sub FontLoad()
    Dim q() As Byte = My.Resources.font
    Dim n As Integer = 0
    '匯入六與七碼字模
    For m As Integer = 0 To 1

```

```

        For k As Integer = 0 To 35
            For j As Integer = 0 To Ph - 1
                For i As Integer = 0 To Pw - 1
                    P(m, k, i, j) = q(n)
                    n += 1
                Next
            Next
        Next
    Next
Next
'匯入六碼變形 69 字模
For k As Integer = 0 To 1
    For j As Integer = 0 To Ph - 1
        For i As Integer = 0 To Pw - 1
            P69(k, i, j) = q(n)
            n += 1
        Next
    Next
Next
Next
End Sub

```

這個副程式是建構二進位檔案的逆運算，將所有資料倒回 P 與 P69 兩個陣列。此時再次執行專案，你就不再需要匯入圖檔，可以直接選擇顯示出各個字模的圖案了！如果是可以辨識非常多種字型與文字的 OCR 軟體，通常資料庫很大，但是車牌辨識的字元與字型有限，我們用這種資源檔案的方式就可以很容易讓程式自動取得需要的基本資料。後續的專案只要保留這個字模檔案載入為資源檔，開程式時用 FontLoad 副程式「解碼」即可。

6-6 字模比對的示範

2

上圖是我用小畫家畫出與字模檔案同樣大小的圖檔，請寫 PictureBox2_Click 的事件副程式載入這個檔案，並選擇六碼車牌的標準 2 字，準備讓此手繪的 2 與字模作符合度的對比，程式碼如下：

```

'載入比對目標
Private Sub PictureBox2_Click(ByVal sender As Object, ByVal e As System.EventArgs) _
    Handles PictureBox2.Click
    If OpenFileDialog1.ShowDialog = Windows.Forms.DialogResult.OK Then
        Dim m As New Bitmap(OpenFileDialog1.FileName)
        PictureBox2.Image = m
        ReDim A(Pw - 1, Ph - 1) '待比對的目標二值化陣列
        For i As Integer = 0 To Pw - 1

```

```

        For j As Integer = 0 To Ph - 1
            Dim C As Color = m.GetPixel(i, j)
            If C.R < 127 Then A(i, j) = 1 '黑點
        Next
    Next
End If
End Sub

```


請注意到，在載入 PictureBox1 與 PictureBox2 兩個影像時，我們已經順便建立好了兩者的二值化陣列 B 與 A，要計算兩者有多像？只要比對它們的黑白點有多符合即可，「**計算符合度**」這個功能按鍵就是想建立一種最簡單的符合度計算方式。原始想法是以字模的總黑點數為分母，這些黑點中與目標陣列的黑點符合的個數為分子，兩者的百分比就是此處定義的符合度了！

但是這種作法是有漏洞的！譬如你要比對的目標其實是 T 時，如果以字模 1 或 I 去比對，符合度也是接近百分百的！又如 P 字用 F 去比對也會「非常符合」！要避免這種狀況就必須有**扣分機制**，就是當字模為白點但目標為黑點時表示「**不符合**」必須扣點。按此邏輯設計之程式碼與執行結果如下：

```

'執行字模比對
Private Sub Button3_Click(ByVal sender As Object, ByVal e As System.EventArgs) _
    Handles Button3.Click
    Dim n0 As Integer = 0 '字模黑點數
    Dim nf As Integer = 0 '符合的黑點數
    For i As Integer = 0 To Pw - 1
        For j As Integer = 0 To Ph - 1
            If B(i, j) = 0 Then Continue For
            n0 += 1 '字模黑點數累計
            If A(i, j) = 1 Then nf += 1 '目標與字模符合點數
        Next
    Next
    Dim pc As Integer = nf * 1000 / n0 '符合點數千分比
    Label1.Text = pc.ToString
End Sub

```

6-7 有關此章內容的討論

本章可以說是在製作車牌辨識的最終字元的標準答案，與符合度的計分方式，這部分與前後兩章都沒有直接關係。前一章我們已經介紹了如何找到車牌字元目標的群組，下一章開始會進一步介紹如何讓它們的形狀轉正到可以和字模比對。建議大家保留此專案當作工具程式，隨時可用於調整字模圖檔重建資料，或比對實測資料中的目標與字模的符合度。

坦白說，我們銷售中的商業軟體，不論是字模建立的方式，或比對計分的演算法都比此地介紹的內容複雜精緻很多！原因主要是為了**強化抗模糊能力**，就是提升當目標扭曲、破碎或變形時的辨識能力！譬如 B 與 8 很像，我們會有額外的加權計分，譬如尖銳轉角的會傾向是 B，圓角的傾向是 8 等等。

這部份我們將不會在本書詳述，原因不完全是為了商業機密，而是那些額外程式太過瑣碎複雜，不太具有教學的價值意義。本書的目的是想讓大家知道最主要的影像辨識實作的核心流程，建立「**我也做得到**」的信心！但如果你想要辨識更準，搞定更多在可正確辨識邊緣的不完美影像，就還需要持續努力，在計分方式等方面作更多嘗試實驗與研究，羅馬不是一天造成的！但是影像辨識的主要梁柱我一定會幫讀者建立好！

完整專案

Public Class Form1

Dim Pw As Integer = 25, Ph As Integer = 50 '字模的寬與高

Dim P(1, 35, Pw - 1, Ph - 1) As Byte '六與七碼車牌所有英數字二值化陣列

Dim P69(1, Pw - 1, Ph - 1) As Byte '變形的六碼車牌 6 與 9 字型

Dim A(Pw - 1, Ph - 1) As Byte, B(Pw - 1, Ph - 1) As Byte '待比對的字模與目標二值化陣列
'用影像載入字模

Private Sub Button1_Click(ByVal sender As Object, ByVal e As System.EventArgs) _
Handles Button1.Click

If FolderBrowserDialog1.ShowDialog = Windows.Forms.DialogResult.OK Then

Dim d As String = FolderBrowserDialog1.SelectedPath

For k As Integer = 0 To 35

'六碼字型

Dim b6 As New Bitmap(d + "\" + Format(k, "00") + ".gif")

For i As Integer = 0 To Pw - 1

For j As Integer = 0 To Ph - 1

Dim C As Color = b6.GetPixel(i, j)

If C.R < 128 Then P(0, k, i, j) = 1 '黑點

Next

Next

'七碼字型

Dim b7 As New Bitmap(d + "\" + Format(k + 36, "00") + ".gif")

For i As Integer = 0 To Pw - 1

For j As Integer = 0 To Ph - 1

Dim C As Color = b7.GetPixel(i, j)

If C.R < 128 Then P(1, k, i, j) = 1 '黑點

Next

Next

Next

For k As Integer = 72 To 73

'六碼變形的 69 字型

Dim b69 As New Bitmap(d + "\" + Format(k, "00") + ".gif")

For i As Integer = 0 To Pw - 1

For j As Integer = 0 To Ph - 1

Dim C As Color = b69.GetPixel(i, j)

If C.R < 128 Then P69(k - 72, i, j) = 1 '黑點

Next

Next

Next

End If

End Sub

'選擇字模

Private Sub ListBox1_SelectedIndexChanged(ByVal sender As Object, ByVal e As System.EventArgs) _
Handles ListBox1.SelectedIndexChanged

Dim knd As Integer = ListBox2.SelectedIndex '六或七碼

If knd < 0 Then Exit Sub '未選擇

Dim sC As Integer = ListBox1.SelectedIndex '選擇的字元

If sC < 0 Then Exit Sub '未選擇

If knd = 1 And sC > 35 Then Exit Sub '七碼車牌字元無變形的 69

ReDim B(Pw - 1, Ph - 1) '待比對的字模二值化陣列

For i As Integer = 0 To Pw - 1

For j As Integer = 0 To Ph - 1

If knd = 0 Then '六碼字型

If sC <= 35 Then

B(i, j) = P(0, sC, i, j)

Else

```

        B(i, j) = P69(sC - 36, i, j)
    End If
    Else '七碼字型
        B(i, j) = P(1, sC, i, j)
    End If
Next
Next
PictureBox1.Image = BWImg(B) '顯示字模影像
End Sub
'建立字模二進位檔案
Private Sub Button2_Click(ByVal sender As Object, ByVal e As EventArgs) _
    Handles Button2.Click
    If SaveFileDialog1.ShowDialog = Windows.Forms.DialogResult.OK Then
        If My.Computer.FileSystem.FileExists(SaveFileDialog1.FileName) Then
            My.Computer.FileSystem.DeleteFile(SaveFileDialog1.FileName)
        End If
        '寫出六與七碼字模
        For m As Integer = 0 To 1
            For k As Integer = 0 To 35
                For j As Integer = 0 To Ph - 1
                    Dim x(Pw - 1) As Byte
                    For i As Integer = 0 To Pw - 1
                        x(i) = P(m, k, i, j)
                    Next
                    My.Computer.FileSystem.WriteAllBytes(SaveFileDialog1.FileName, x, True)
                Next
            Next
        Next
        '寫出六碼變形 69 字模
        For k As Integer = 0 To 1
            For j As Integer = 0 To Ph - 1
                Dim x(Pw - 1) As Byte
                For i As Integer = 0 To Pw - 1
                    x(i) = P69(k, i, j)
                Next
                My.Computer.FileSystem.WriteAllBytes(SaveFileDialog1.FileName, x, True)
            Next
        Next
    End If
End Sub
'啟動程式載入字模
Private Sub Form1_Load(ByVal sender As Object, ByVal e As System.EventArgs) Handles MyBase.Load
    FontLoad()
End Sub
Private Sub FontLoad()
    Dim q() As Byte = My.Resources.font
    Dim n As Integer = 0
    '匯入六與七碼字模
    For m As Integer = 0 To 1
        For k As Integer = 0 To 35
            For j As Integer = 0 To Ph - 1
                For i As Integer = 0 To Pw - 1
                    P(m, k, i, j) = q(n)
                    n += 1
                Next
            Next
        Next
    Next
Next
Next
Next

```

```

'匯入六碼變形 69 字模
For k As Integer = 0 To 1
    For j As Integer = 0 To Ph - 1
        For i As Integer = 0 To Pw - 1
            P69(k, i, j) = q(n)
            n += 1
        Next
    Next
Next
End Sub
'載入比對目標
Private Sub PictureBox2_Click(ByVal sender As Object, ByVal e As System.EventArgs) _
    Handles PictureBox2.Click
    If OpenFileDialog1.ShowDialog = Windows.Forms.DialogResult.OK Then
        Dim m As New Bitmap(OpenFileDialog1.FileName)
        PictureBox2.Image = m
        ReDim A(Pw - 1, Ph - 1) '待比對的目標二值化陣列
        For i As Integer = 0 To Pw - 1
            For j As Integer = 0 To Ph - 1
                Dim C As Color = m.GetPixel(i, j)
                If C.R < 127 Then A(i, j) = 1 '黑點
            Next
        Next
    End If
End Sub
'執行字模比對
Private Sub Button3_Click(ByVal sender As Object, ByVal e As System.EventArgs) _
    Handles Button3.Click
    Dim n0 As Integer = 0 '字模黑點數
    Dim nf As Integer = 0 '符合的黑點數
    For i As Integer = 0 To Pw - 1
        For j As Integer = 0 To Ph - 1
            If B(i, j) = 0 Then
                If A(i, j) = 1 Then nf -= 1 '目標與字模不符合扣分
            Else
                n0 += 1 '字模黑點數累計
                If A(i, j) = 1 Then nf += 1 '目標與字模符合加分
            End If
        Next
    Next
    Dim pc As Integer = nf * 1000 / n0 '符合點數千分比
    Label1.Text = pc.ToString
End Sub
End Class

```